Procedures

CSE 410, Spring 2009

Computer Systems

http://www.cs.washington.edu/410

Readings and References

- Reading
  - » Section 2.8, Supporting Procedures in Computer Hardware
  - » Section B.5, Memory Usage
  - » Section B.6, Procedure Call Convention

Instructions and Data flow



Layout of program memory



Not to Scale!

Why use procedures?

- So far, our program is just one long run of instructions
- We can do a lot this way, but the program rapidly gets too large to handle easily
- Procedures allow the programmer to organize the code into logical units

What does a procedure do for us?

- A procedure provides a well defined and reusable interface to a particular capability
  - » entry, exit, parameters clearly identified
- Reduces the level of detail the programmer needs to know to accomplish a task
- Caller can ignore the internals of a function
  - » messy details can be hidden from innocent eyes
  - » internals can change without affecting caller

How does a procedure call work?

1. set up parameters
2. transfer to procedure
3. acquire storage resources
4. do the desired function
5. make result available to caller
6. release storage resources
7. return to point of call

Calling conventions

- The details of how you implement the steps for using a procedure are governed by the *calling conventions* being used
- There is much variation in conventions
  - » which causes much programmer pain
- Understand the calling conventions of the system you are writing for
  - » o32, n32, n64, P&H, cse410, ...

1. Set up parameters

- The registers are one obvious place to put parameters for a procedure to read
  - » very fast and easily referenced
- Many procedures have 4 or less arguments
  - » MIPS: $a0, $a1, $a2, $a3 are used for arguments
- … but some procedures have more
  - » we don't want to use up all the registers
  - » so we use memory to store the rest

The Stack

- Stack pointer ($sp) points to the “top” value on the stack (ie, the lowest address in use)
- MIPS has no “push” or “pop” instructions
  - » we adjust the stack pointer directly
- Stack grows downward towards zero
  - » subu $sp, $sp, xx : make room for more data
  - » addu $sp, $sp, xx : release space on the stack
  - » note that both subu and addu become addiu

Dynamic storage on the stack

```
...
jal main
```

```
main:
  subu $sp,$sp,8
  ...
```



$sp `0x7fffedf8`

$sp `0x7fffedf0`

Layout of stack frame



Argument build area

- Some MIPS calling conventions require that caller reserve stack space for all arguments
  - » 16 bytes (4 words) left empty to mirror **`$a0-$a3`**
- Other calling conventions require that caller reserve stack space only for arguments that do not fit in **`$a0 - $a3`**
  - » so argument build area is only present if some arguments didn't fit in 4 registers

Agreement

- A procedure and all of the programs that call it must agree on the calling convention
- This is one reason why changing the calling convention for system libraries is a big deal
- We will use
  - » caller reserves stack space for all arguments
  - » 16 bytes (4 words) left empty to mirror **`$a0-$a3`**

2. Transfer to procedure



Jump and link

- Jump
  - » can take you anywhere within the currently active 256 MB segment
- Link
  - » store return address in $ra
  - » note: this overwrites current value of $ra

3. Acquire storage resources



3a. Saved registers

- There is only one set of registers
  - » If called procedure unexpectedly overwrites them, caller will be surprised and distressed
- Another agreement
  - » called procedure can change $a0-$a3, $v0-$v1, $t0-$t9 without restoring original values
  - » called procedure must save and restore value of any other register it wants to use

Register numbers and names

number	*name*	*usage*
0	**zero**	always returns 0
1	**at**	reserved for use as assembler temporary
2-3	**v0, v1**	values returned by procedures
4-7	**a0-a3**	first few procedure arguments
8-15, 24, 25	**t0-t9**	temps - can use without saving
16-23	**s0-s7**	temps - must save before using
26,27	**k0, k1**	reserved for kernel use - may change at any time
28	**gp**	global pointer
29	**sp**	stack pointer
30	**fp** or **s8**	frame pointer
31	**ra**	return address from procedure

3b. Local variables

- If the called procedure needs to store values in memory while it is working, space must be reserved on the stack for them
- Debugging note
  - » compiler can often optimize so that all variables fit in registers and are never stored in memory
  - » so a memory dump may not contain all values
  - » use switches to turn off optimization (but …)

3c. Argument build area

- Our convention is
  - » caller reserves stack space for all arguments
  - » 16 bytes (4 words) left empty to mirror **`$a0-$a3`**
- If your procedure does more than one call to other procedures, then ...
  - » the argument build area must be large enough for the largest set of arguments

Using the stack pointer

- Adjust it once on entry, once on exit
  - » Initial adjustment should include all the space you will need in this procedure
- Remember that a word is 4 bytes
  - » so expect to see references like **`8($sp), 20($sp)`**
- Keep stack pointer double word aligned
  - » adjust by multiples of 8

4. Do the desired function

- You have saved the values of the registers that must be preserved across the call
- The arguments are in $a0 - $a3 or on the stack
- The stack pointer points to the end of your stack frame
- Let ‘er rip
  - » signal processing, image filter, encryption, …

5. Make result available to caller

- Registers $v0 and $v1 are available for this
- Most procedures put a 32-bit value in $v0
- Returning the address of a variable?
  - » be very careful!
  - » your portion of the stack is invalid as soon as you return
  - » the object must be allocated in caller's part of stack (or somewhere further back), or globally allocated (heap or static storage)

6. Return storage resources



7. Return to point of call

- Jump through register
- The address of the instruction following the jump and link was put in $ra when we were called (the "link" in jump and link)
- We have carefully preserved $ra while the procedure was executing
- So, "**`jr $ra`**" takes us right back to caller

CSE 410 Calling Conventions

- Argument build area
  - » caller reserves stack space for all arguments
  - » 16 bytes (4 words) left empty to mirror $a0-$a3
- Called procedure adjusts stack pointer once on entry, once on exit, in units of 8 bytes
- Register usage in functions
  - » not required to save and restore $t0-$t9, $a0-$a3
  - » must save and restore $s0-$s8, $ra if changed
  - » function results returned in $v0, $v1

Leaf procedures

- A leaf procedure is one that does not call another procedure
- Relatively simple register usage since the procedure doesn't call anyone else
- Little or no memory access requirements because you are not saving and restoring as many registers from the stack

Non-leaf procedure

- A non-leaf procedure is one that calls another procedure
- You must save at least register **`$ra`**, since that register is overwritten by the **`jal`** when you call another procedure

Calling tree



Layout of stack frame (little leaf)




```
procC:
   subu $sp,$sp,xx
   ...
```

Layout of stack frame (big leaf)




```
procC:
  subu $sp,$sp,xx
  ...
```

Layout of stack frame (non-leaf)



Little leaf example - swap.c

```
/* Swap two integer array elements */

void swap(int a[], int i, int j)
{
        int T;
        T = a[i];
        a[i] = a[j];
        a[j] = T;
}
```

Little leaf example - swap.s

```
swap:
    sll     $a1,$a1,2          # $a1 = 4*i
    addu    $a1,$a1,$a0        # $a1 = addr(a[i])
    lw      $v1,0($a1)         # $v1 = a[i]
    sll     $a2,$a2,2          # $a2 = 4*j
    addu    $a2,$a2,$a0        # $a2 = addr(a[j])
    lw      $v0,0($a2)         # $v0 = a[j]
    sw      $v0,0($a1)         # a[i] = old a[j]
    sw      $v1,0($a2)         # a[j] = old a[i]
    j       $ra                # return
```

Non-leaf example - QuickSort.c

```
void QuickSort(int a[], int lo0, int hi0)
{
        int lo = lo0;
        int hi = hi0;
        int mid;

        if ( hi0 > lo0)
        {
        ...
```

Non-leaf example - QuickSort.s

```
QuickSort:
   subu    $sp,$sp,48          # create stack frame
   sw      $ra,40($sp)         #
   sw      $s5,36($sp)         #
   sw      $s4,32($sp)         #
   sw      $s3,28($sp)         #
   sw      $s2,24($sp)         #
   sw      $s1,20($sp)         #
   sw      $s0,16($sp)         #
   move    $s3,$a0             # $s3 = address(a)
   move    $s5,$a1             # $s5 = lo0
   ...
```

Layout of QuickSort stack frame



$ra - Return Address

- Return address register
  - » written with jal, jalr instructions
  - » must be saved if procedure calls another

```
QuickSort:
  subu    $sp,$sp,48    # create stack frame
  sw      $ra,40($sp)   #
  . . .
  lw      $ra,40($sp)   # restore from stack …
  addu    $sp,$sp,48    #
  j       $ra           # return
```

$fp - Frame Pointer

- Frame pointer points to the largest address in the stack frame
- Stack pointer points to the smallest address in the stack frame
  - » no advantage to $fp if $sp does not change during procedure's execution
- Consider $fp to be $s8
  - » save and restore required if you use it

Layout of stack frame (with $fp)



$s0-$s7 - Save and Restore

- These registers are available for unlimited use
- Must save immediately on procedure entry and restore just before procedure exit if you are going to use them
- As a result of this convention, the registers will have the same values after a procedure call as they had before

$t0-$t9 - Temporary registers

- Use however you like
- No save and restore required or expected
- As a result of this convention, the registers have no guaranteed values when you get back from calling another procedure

$a0-$a3 , $v0-$v1 - Args/Return

- The argument registers can be changed in a procedure without restriction
- No guarantee that they will be the same upon return from a called procedure
- The result registers will contain whatever the function prototype says they will
  - » undefined value in $v1 if not used for return

Some Perspective

- These calling conventions can look very complex
  - » but partly that's just appalling documentation
  - » and the inclusion of debugging conventions
- Most functions that you may write in assembler for tuning reasons will be leaf functions
  - » the declaration of such a function is very simple